Exploring black hole spacetimes with SageManifolds

Éric Gourgoulhon

Laboratoire Univers et Théories (LUTH)
CNRS / Observatoire de Paris / Université Paris Diderot
92190 Meudon, France

http://luth.obspm.fr/~luthier/gourgoulhon/

joint work with

Pablo Angulo, Michał Bejger, Marco Mancini, Travis Scrimshaw (SageManifolds)

Irina Aref'eva, Anastasia Golubtsova (gravity/gauge duality)

21st International Conference on General Relativity and Gravitation
New York
10-15 July 2016

SageMath in a few words

- SageMath (formerly Sage) is a **free open-source** mathematics software system [http://www.sagemath.org/]
- It is based on the **Python** programming language
- It can be
  - freely downloaded and installed on one's computer
  - used online by opening a (free) account on the SageMathCloud: https://cloud.sagemath.com/
- William Stein (Univ. of Washington) created SageMath in 2005; since then, ~**100 developers** (mostly mathematicians) have joined the SageMath team
- SageMath is now supported by European Union via the open-math project OpenDreamKit (2015-2019, within the *Horizon 2020* program)

The mission

Create a viable free open source alternative to Magma, Maple, Mathematica and Matlab.

The SageManifolds project

http://sagemanifolds.obspm.fr/

Aim

Implement smooth manifolds of arbitrary dimension in SageMath and tensor calculus on them

In particular, one should be able

- to introduce an arbitrary number of coordinate charts on a given manifold, with the relevant transition maps
- to express tensor fields in terms of their components in various (possibly non-coordinate) vector frames
- to deal with tensor fields on non-parallelizable manifolds (i.e. without any global vector frame)

Some mathematical structures in SageManifolds

The implementation of SageManifolds follows SageMath's parent/element framework

Scalar fields

- Given an open subset U of manifold M, a **scalar field on** U is a smooth map $f : U \to \mathbb{R}$
  f maps *points*, not *coordinates*, to real numbers $\Longrightarrow$ f has different **coordinate representations** in different charts defined on U.
- The set $C^\infty(U)$ of scalar fields on U is a **commutative algebra over** $\mathbb{R}$
  $C^\infty(U)$ is the parent of f.

Some mathematical structures in SageManifolds

The implementation of SageManifolds follows SageMath's parent/element framework

Scalar fields

- Given an open subset U of manifold M, a **scalar field on** U is a smooth map $f : U \to \mathbb{R}$
  f maps *points*, not *coordinates*, to real numbers $\Longrightarrow$ f has different **coordinate representations** in different charts defined on U.
- The set $C^\infty(U)$ of scalar fields on U is a **commutative algebra over** $\mathbb{R}$
  $C^\infty(U)$ is the parent of f.

Vector fields

The set $\mathcal{X}(U)$ of smooth vector fields defined on U is a **module over the scalar field algebra** $C^\infty(U)$. $\mathcal{X}(U)$ is the parent of vector fields on U.

$\mathcal{X}(U)$ is a **free module** $\iff$ $\mathcal{X}(U)$ admits a basis
$\iff$ U admits a global vector frame
$\iff$ U is parallelizable
$\iff$ U's tangent bundle is trivial: $TU \simeq U \times \mathbb{R}^n$

Example: black branes in 5-dim Lifshitz-like spacetimes

A full example with SageManifolds running in a Jupyter notebook



Motivation: study of the anisotropic phase of the quark-gluon plasma via the gravity/gauge duality
cf. arXiv:1601.06046

Get/read the worksheet at
http://nbviewer.ipython.org/github/sagemanifolds/SageManifolds/blob/master/Worksheets/v0.9/SM_black_brane.ipynb

Black branes in Lifshitz-like spacetimes

This Jupyter/SageMath worksheet implements some computations of the article

- I. Ya. Aref'eva, A. A. Golubtsova & E. Gourgoulhon: *Analytic black branes in Lifshitz-like backgrounds and thermalization*, arXiv:1601.06046

These computations are based on SageManifolds (v0.9)

The worksheet file (ipynb format) can be downloaded from here.

First we set up the notebook to display mathematical objects using LaTeX formatting:

In [1]:
```
%display latex
```

Space manifold

Let us declare M as a 5-dimensional manifold:

In [2]:
```
M = Manifold(5, 'M')
print(M)
```

```
5-dimensional differentiable manifold M
```

We introduce a coordinate system on M:

In [3]:
```
X.<t,x,y1,y2,r> = M.chart('t x y1:y_1 y2:y_2 r')
X
```

Out[3]: $(M, (t, x, y_1, y_2, r))$

Next, we define the metric tensor, which depends on some real number ν and some arbitary function f:

In [4]:
```
g = M.lorentzian_metric('g')
var('nu', latex_name=r'\nu', domain='real')
ff = function('f')(r)
g[0,0] = -ff*exp(2*nu*r)
g[1,1] = exp(2*nu*r)
g[2,2] = exp(2*r)
g[3,3] = exp(2*r)
g[4,4] = 1/ff
g.display()
```

Out[4]:

$$g = -e^{(2\,\nu r)} f(r)\,\mathrm{d}t \otimes \mathrm{d}t + e^{(2\,\nu r)}\mathrm{d}x \otimes \mathrm{d}x + e^{(2\,r)}\mathrm{d}y_1 \otimes \mathrm{d}y_1 + e^{(2\,r)}\mathrm{d}y_2 \otimes \mathrm{d}y_2 + \frac{1}{f(r)}\mathrm{d}r \otimes \mathrm{d}r$$

If $f(r) = 1$, this is the metric of a Lifshitz spacetime; if, in addition $\nu = 1$, (M, g) is a Poincaré patch of AdS_5.

Curvature

The Riemann tensor is

In [5]:
```
Riem = g.riemann()
print(Riem)
```

```
Tensor field Riem(g) of type (1,3) on the 5-dimensional differentiable m
anifold M
```

In [6]:
```
Riem.display_comp(only_nonredundant=True)
```

Out[6]:

$$\mathrm{Riem}(g)^{t}{}_{x\,t\,x} = -\nu^2 e^{(2\,\nu r)} f(r) - \frac{1}{2}\,\nu e^{(2\,\nu r)} \frac{\partial f}{\partial r}$$

$$\mathrm{Riem}(g)^{t}{}_{y_1\,t\,y_1} = -\nu e^{(2\,r)} f(r) - \frac{1}{2}\,e^{(2\,r)} \frac{\partial f}{\partial r}$$

$$\mathrm{Riem}(g)^{t}{}_{y_2\,t\,y_2} = -\nu e^{(2\,r)} f(r) - \frac{1}{2}\,e^{(2\,r)} \frac{\partial f}{\partial r}$$

$$\mathrm{Riem}(g)^{t}{}_{r\,t\,r} = -\frac{2\,\nu^2 f(r) + 3\,\nu \frac{\partial f}{\partial r} + \frac{\partial^2 f}{\partial r^2}}{2\,f(r)}$$

$$\mathrm{Riem}(g)^{x}{}_{t\,t\,x} = -\nu^2 e^{(2\,\nu r)} f(r)^2 - \frac{1}{2}\,\nu e^{(2\,\nu r)} f(r) \frac{\partial f}{\partial r}$$

$$\mathrm{Riem}(g)^{x}{}_{y_1\,x\,y_1} = -\nu e^{(2\,r)} f(r)$$

$$\mathrm{Riem}(g)^{x}{}_{y_2\,x\,y_2} = -\nu e^{(2\,r)} f(r)$$

$$\mathrm{Riem}(g)^{x}{}_{r\,x\,r} = -\frac{2\,\nu^2 f(r) + \nu \frac{\partial f}{\partial r}}{2\,f(r)}$$

$$
\begin{aligned}
\mathrm{Riem}(g)^{y_1}{}_{t\,t\,y_1} &= -\nu e^{(2\,\nu r)} f(r)^2 - \frac{1}{2}\, e^{(2\,\nu r)} f(r)\, \frac{\partial f}{\partial r} \\
\mathrm{Riem}(g)^{y_1}{}_{x\,x\,y_1} &= \nu e^{(2\,\nu r)} f(r) \\
\mathrm{Riem}(g)^{y_1}{}_{y_2\,y_1\,y_2} &= -e^{(2\,r)} f(r) \\
\mathrm{Riem}(g)^{y_1}{}_{r\,y_1\,r} &= -\frac{2\,f(r) + \frac{\partial f}{\partial r}}{2\,f(r)} \\
\mathrm{Riem}(g)^{y_2}{}_{t\,t\,y_2} &= -\nu e^{(2\,\nu r)} f(r)^2 - \frac{1}{2}\, e^{(2\,\nu r)} f(r)\, \frac{\partial f}{\partial r} \\
\mathrm{Riem}(g)^{y_2}{}_{x\,x\,y_2} &= \nu e^{(2\,\nu r)} f(r) \\
\mathrm{Riem}(g)^{y_2}{}_{y_1\,y_1\,y_2} &= e^{(2\,r)} f(r) \\
\mathrm{Riem}(g)^{y_2}{}_{r\,y_2\,r} &= -\frac{2\,f(r) + \frac{\partial f}{\partial r}}{2\,f(r)} \\
\mathrm{Riem}(g)^{r}{}_{t\,t\,r} &= -\nu^2 e^{(2\,\nu r)} f(r)^2 - \frac{3}{2}\,\nu e^{(2\,\nu r)} f(r)\, \frac{\partial f}{\partial r} - \frac{1}{2}\, e^{(2\,\nu r)} f(r)\, \frac{\partial^2 f}{\partial r^2} \\
\mathrm{Riem}(g)^{r}{}_{x\,x\,r} &= \nu^2 e^{(2\,\nu r)} f(r) + \frac{1}{2}\,\nu e^{(2\,\nu r)}\, \frac{\partial f}{\partial r} \\
\mathrm{Riem}(g)^{r}{}_{y_1\,y_1\,r} &= e^{(2\,r)} f(r) + \frac{1}{2}\, e^{(2\,r)}\, \frac{\partial f}{\partial r} \\
\mathrm{Riem}(g)^{r}{}_{y_2\,y_2\,r} &= e^{(2\,r)} f(r) + \frac{1}{2}\, e^{(2\,r)}\, \frac{\partial f}{\partial r}
\end{aligned}
$$

The Ricci tensor:

In [7]:
```
Ric = g.ricci()
print(Ric)
```

```
Field of symmetric bilinear forms Ric(g) on the 5-dimensional differenti
able manifold M
```

In [8]:
```
Ric.display()
```

Out[8]:

$$\mathrm{Ric}\left(g\right) = \left(2\left(\nu^{2}+\nu\right)e^{\left(2\,\nu r\right)}f(r)^{2} + \left(2\,\nu+1\right)e^{\left(2\,\nu r\right)}f\left(r\right)\frac{\partial f}{\partial r} + \frac{1}{2}\,e^{\left(2\,\nu r\right)}f\left(r\right)\frac{\partial^{2} f}{\partial r^{2}}\right)\mathrm{d}t \otimes \mathrm{d}t + \left(-2\left(\nu^{2}+\nu\right)e^{\left(2\,\nu r\right)}f\left(r\right) - \nu e^{\left(2\,\nu r\right)}\frac{\partial f}{\partial r}\right)\mathrm{d}x \otimes \mathrm{d}x + \left(-2\left(\nu+1\right)e^{\left(2\,r\right)}f\left(r\right) - e^{\left(2\,r\right)}\frac{\partial f}{\partial r}\right)\mathrm{d}y_{1} \otimes \mathrm{d}y_{1} + \left(-2\left(\nu+1\right)e^{\left(2\,r\right)}f\left(r\right) - e^{\left(2\,r\right)}\frac{\partial f}{\partial r}\right)\mathrm{d}y_{2} \otimes \mathrm{d}y_{2} + \left(-\frac{4\left(\nu^{2}+1\right)f\left(r\right) + 2\left(2\,\nu+1\right)\frac{\partial f}{\partial r} + \frac{\partial^{2} f}{\partial r^{2}}}{2\,f\left(r\right)}\right)\mathrm{d}r \otimes \mathrm{d}r$$

```
In [9]: Ric.display_comp()
```

Out[9]:

$$\mathrm{Ric}(g)_{tt} = 2\left(\nu^2+\nu\right)e^{(2\nu r)}f(r)^2 + (2\nu+1)e^{(2\nu r)}f(r)\frac{\partial f}{\partial r} + \frac{1}{2}e^{(2\nu r)}f(r)\frac{\partial^2 f}{\partial r^2}$$

$$\mathrm{Ric}(g)_{xx} = -2\left(\nu^2+\nu\right)e^{(2\nu r)}f(r) - \nu e^{(2\nu r)}\frac{\partial f}{\partial r}$$

$$\mathrm{Ric}(g)_{y_1 y_1} = -2(\nu+1)e^{(2r)}f(r) - e^{(2r)}\frac{\partial f}{\partial r}$$

$$\mathrm{Ric}(g)_{y_2 y_2} = -2(\nu+1)e^{(2r)}f(r) - e^{(2r)}\frac{\partial f}{\partial r}$$

$$\mathrm{Ric}(g)_{rr} = -\frac{4\left(\nu^2+1\right)f(r)+2(2\nu+1)\frac{\partial f}{\partial r}+\frac{\partial^2 f}{\partial r^2}}{2f(r)}$$

The Ricci scalar:

```
In [10]: Rscal = g.ricci_scalar()
         print(Rscal)
```

```
Scalar field r(g) on the 5-dimensional differentiable manifold M
```

```
In [11]: Rscal.display()
```

Out[11]:

$$\begin{array}{llcl} \mathrm{r}(g): & M & \longrightarrow & \mathbb{R} \\ & (t,x,y_1,y_2,r) & \longmapsto & -2\left(3\nu^2+4\nu+3\right)f(r) - (5\nu+4)\frac{\partial f}{\partial r} - \frac{\partial^2 f}{\partial r^2} \end{array}$$

Source model

Let us consider a model based on the following action, involving a cosmological constant $\bar{\Lambda} = -\Lambda/2$ with $\Lambda > 0$, a dilaton scalar field ϕ and a Maxwell 2-form F:

$$S = \int \left(R(g) + \Lambda - \frac{1}{2}\nabla_m\phi\nabla^m\phi - \frac{1}{4}e^{\lambda\phi}F_{mn}F^{mn} \right) \sqrt{-g}\, \mathrm{d}^5 x \qquad (1)$$

where $R(g)$ is the Ricci scalar of metric g and λ is the dilatonic coupling constant.

The dilaton scalar field

We consider the following ansatz for the dilaton scalar field ϕ:

$$\phi = \frac{1}{\lambda}(4r + \ln\mu) \iff e^{\lambda\phi} = \mu e^{4r},$$

where μ is a constant.

In [12]:
```
var('mu', latex_name=r'\mu', domain='real')
var('lamb', latex_name=r'\lambda', domain='real')
phi = M.scalar_field({X: (4*r + ln(mu))/lamb},
                     name='phi', latex_name=r'\phi')
phi.display()
```

Out[12]:

$$\begin{array}{llcl} \phi: & M & \longrightarrow & \mathbb{R} \\ & (t, x, y_1, y_2, r) & \longmapsto & \frac{4\,r + \log(\mu)}{\lambda} \end{array}$$

The 1-form $\mathrm{d}\phi$ is

In [13]:
```
dphi = phi.differential()
print(dphi)
```

```
1-form dphi on the 5-dimensional differentiable manifold M
```

In [14]:
```
dphi.display()
```

Out[14]:

$$\mathrm{d}\phi = \frac{4}{\lambda}\mathrm{d}r$$

In [15]:
```
dphi[:]  # all the components in the default frame
```

Out[15]:

$$\left[0, 0, 0, 0, \frac{4}{\lambda}\right]$$

The 2-form field

We consider the following ansatz for F:

$$F = \frac{1}{2} q \, \mathrm{d}y_1 \wedge \mathrm{d}y_2,$$

where q is a constant. Let us first get the 1-forms $\mathrm{d}y_1$ and $\mathrm{d}y_2$:

In [16]:
```
X.coframe()
```

Out[16]:

$$(M, (\mathrm{d}t, \mathrm{d}x, \mathrm{d}y_1, \mathrm{d}y_2, \mathrm{d}r))$$

In [17]:
```
dy1 = X.coframe()[2]
dy2 = X.coframe()[3]
```

Then we can form F according to the above ansatz:

In [18]:
```
var('q', domain='real')
F = q/2 * dy1.wedge(dy2)
F.set_name('F')
print(F)
F.display()
```

```
2-form F on the 5-dimensional differentiable manifold M
```

Out[18]:

$$F = \frac{1}{2}\, q \mathrm{d}y_1 \wedge \mathrm{d}y_2$$

By construction, the 2-form F is closed (since q is constant):

In [19]:
```
print(F.exterior_der())
```

```
3-form dF on the 5-dimensional differentiable manifold M
```

In [20]:
```
F.exterior_der().display()
```

Out[20]:

$$\mathrm{d}F = 0$$

Let us evaluate the square $F_{mn}F^{mn}$ of F:

In [21]:
```
Fu = F.up(g)
print(Fu)
```

```
Tensor field of type (2,0) on the 5-dimensional differentiable manifold
M
```

In [22]:
```
F2 = F['_{mn}']*Fu['^{mn}']  # using LaTeX notations for contraction
print(F2)
F2.display()
```

```
Scalar field on the 5-dimensional differentiable manifold M
```

Out[22]:

$$\begin{array}{llcl} M & \longrightarrow & \mathbb{R} \\ (t,x,y_1,y_2,r) & \longmapsto & \frac{1}{2}\,q^2 e^{(-4\,r)} \end{array}$$

We shall also need the tensor $\mathcal{F}_{mn} := F_{mp}F_n{}^p$:

In [23]:
```
FF = F['_mp'] * F.up(g,1)['^p_n']
print(FF)
FF.display()
```

```
Tensor field of type (0,2) on the 5-dimensional differentiable manifold
M
```

Out[23]:

$$\frac{1}{4}\,q^2 e^{(-2\,r)}\mathrm{d}y_1 \otimes \mathrm{d}y_1 + \frac{1}{4}\,q^2 e^{(-2\,r)}\mathrm{d}y_2 \otimes \mathrm{d}y_2$$

The tensor field $\mathcal{F}$ is symmetric:

```
In [24]: FF == FF.symmetrize()
```

Out[24]: True

Therefore, from now on, we set

```
In [25]: FF = FF.symmetrize()
```

Einstein equation

Let us first introduce the cosmological constant:

In [26]:
```
var('Lamb', latex_name=r'\Lambda', domain='real')
```

Out[26]: Λ

From the action (1), the field equation for the metric g is

$$R_{mn} + \frac{\Lambda}{3} g - \frac{1}{2} \partial_m \phi \partial_n \phi - \frac{1}{2} e^{\lambda\phi} F_{mp} F_n{}^p + \frac{1}{12} e^{\lambda\phi} F_{rs} F^{rs} \, g_{mn} = 0$$

We write it as

```
EE == 0
```

with EE defined by

In [27]:
```
EE = Ric + Lamb/3*g - 1/2* (dphi*dphi) - 1/2*exp(lamb*phi)*FF \
     + 1/12*exp(lamb*phi)*F2*g
EE.set_name('E')
print(EE)
```

```
Field of symmetric bilinear forms E on the 5-dimensional differentiable
manifold M
```

In [28]:
```
EE.display_comp(only_nonredundant=True)
```

Out[28]:

$$E_{t\,t} = 2\left(\nu^2+\nu\right)e^{(2\,\nu r)}f(r)^2 + (2\,\nu+1)e^{(2\,\nu r)}f(r)\frac{\partial f}{\partial r} - \frac{1}{24}\left(\mu q^2 + 8\,\Lambda\right)e^{(2\,\nu r)}f(r$$

$$e^{(2\,\nu r)}f(r)\frac{\partial^2 f}{\partial r^2}$$

$$E_{x\,x} = -2\left(\nu^2+\nu\right)e^{(2\,\nu r)}f(r) - \nu e^{(2\,\nu r)}\frac{\partial f}{\partial r} + \frac{1}{24}\left(\mu q^2 + 8\,\Lambda\right)e^{(2\,\nu r)}$$

$$E_{y_1\,y_1} = -2\,(\nu+1)e^{(2\,r)}f(r) - \frac{1}{12}\left(\mu q^2 - 4\,\Lambda\right)e^{(2\,r)} - e^{(2\,r)}\frac{\partial f}{\partial r}$$

$$E_{y_2\,y_2} = -2\,(\nu+1)e^{(2\,r)}f(r) - \frac{1}{12}\left(\mu q^2 - 4\,\Lambda\right)e^{(2\,r)} - e^{(2\,r)}\frac{\partial f}{\partial r}$$

$$E_{r\,r} = \frac{\lambda^2\mu q^2 + 8\,\Lambda\lambda^2 - 12\,\lambda^2\frac{\partial^2 f}{\partial r^2} - 48\left(\lambda^2\nu^2+\lambda^2+4\right)f(r) - 24\left(2\,\lambda^2\nu+\lambda^2\right)\frac{\partial f}{\partial r}}{24\,\lambda^2 f(r)}$$

We note that EE==0 leads to only 4 independent equations:

In [29]:
```
eq0 = EE[0,0]/exp(2*nu*r)
eq0
```

Out[29]:
$$2\left(\nu^2+\nu\right)f(r)^2+(2\,\nu+1)f\left(r\right)\frac{\partial f}{\partial r}-\frac{1}{24}\left(\mu q^2+8\,\Lambda\right)f\left(r\right)+\frac{1}{2}f\left(r\right)\frac{\partial^2 f}{\partial r^2}$$

In [30]:
```
eq1 = EE[1,1]/exp(2*nu*r)
eq1
```

Out[30]:
$$\frac{1}{24}\,\mu q^2-2\left(\nu^2+\nu\right)f\left(r\right)-\nu\frac{\partial f}{\partial r}+\frac{1}{3}\,\Lambda$$

In [31]:
```
eq2 = EE[2,2]/exp(2*r)
eq2
```

Out[31]:
$$-\frac{1}{12}\,\mu q^2-2\,(\nu+1)f\left(r\right)+\frac{1}{3}\,\Lambda-\frac{\partial f}{\partial r}$$

In [32]:
```
eq3 = EE[4,4]*lamb^2*f(r)
eq3
```

Out[32]:
$$\frac{1}{24}\,\lambda^2\mu q^2+\frac{1}{3}\,\Lambda\lambda^2-\frac{1}{2}\,\lambda^2\frac{\partial^2 f}{\partial r^2}-2\,\left(\lambda^2\nu^2+\lambda^2+4\right)f\left(r\right)-\left(2\,\lambda^2\nu+\lambda^2\right)\frac{\partial f}{\partial r}$$

Dilaton field equation

First we evaluate $\nabla_m \nabla^m \phi$:

In [33]:
```
nab = g.connection()
print(nab)
```

```
Levi-Civita connection nabla_g associated with the Lorentzian metric g o
n the 5-dimensional differentiable manifold M
```

In [34]:
```
box_phi = nab(nab(phi).up(g)).trace()
print(box_phi)
box_phi.display()
```

```
Scalar field on the 5-dimensional differentiable manifold M
```

Out[34]:

$$\begin{array}{llcl} & M & \longrightarrow & \mathbb{R} \\ & (t, x, y_1, y_2, r) & \longmapsto & \frac{4\left(2\,(\nu+1)f(r)+\frac{\partial f}{\partial r}\right)}{\lambda} \end{array}$$

From the action (1), the field equation for ϕ is

$$\nabla_m \nabla^m \phi = \frac{\lambda}{4} e^{\lambda\phi} F_{mn} F^{mn}$$

We write it as

```
DE == 0
```

with DE defined by

In [35]:
```
DE = box_phi - lamb/4*exp(lamb*phi) * F2
print(DE)
DE.display()
```

```
Scalar field on the 5-dimensional differentiable manifold M
```

Out[35]:

$$\begin{array}{llll} M & \longrightarrow & \mathbb{R} \\ (t, x, y_1, y_2, r) & \longmapsto & -\frac{\lambda^2 \mu q^2 - 64\,(\nu+1) f(r) - 32\,\frac{\partial f}{\partial r}}{8\,\lambda} \end{array}$$

Hence the dilaton field equation provides a fourth equation:

In [36]:
```
eq4 = DE.coord_function()*lamb
eq4
```

Out[36]:

$$-\frac{1}{8}\,\lambda^2 \mu q^2 + 8\,(\nu+1) f(r) + 4\,\frac{\partial f}{\partial r}$$

Maxwell equation

From the action (1), the field equation for F is

$$\nabla_m \left(e^{\lambda\phi} F^{mn}\right) = 0$$

We write it as

```
ME == 0
```

with ME defined by

In [37]:
```
ME = nab(exp(lamb*phi)*Fu).trace(0,2)
print(ME)
ME.display()
```

```
Vector field on the 5-dimensional differentiable manifold M
```

Out[37]: 0

We get identically zero; indeed the tensor $\nabla_p(e^{\lambda\phi} F^{mn})$ has a vanishing trace, as we can check:

In [38]:
```
nab(exp(lamb*phi)*Fu).display()
```

Out[38]:

$$\mu q \frac{\partial}{\partial y_1} \otimes \frac{\partial}{\partial y_2} \otimes \mathrm{d}r - \frac{1}{2}\mu q e^{(2r)} f(r) \frac{\partial}{\partial y_1} \otimes \frac{\partial}{\partial r} \otimes \mathrm{d}y_2 - \mu q \frac{\partial}{\partial y_2} \otimes \frac{\partial}{\partial y_1} \otimes \mathrm{d}r + \frac{1}{2} \mu q e^{(2r)} f(r) \frac{\partial}{\partial y_2} \otimes \frac{\partial}{\partial r} \otimes \mathrm{d}y_1 + \frac{1}{2}\mu q e^{(2r)} f(r) \frac{\partial}{\partial r} \otimes \frac{\partial}{\partial y_1} \otimes \mathrm{d}y_2 - \frac{1}{2}\mu q e^{(2r)} f(r) \frac{\partial}{\partial r}$$

Solving the field equations

The system to solve is

In [39]:
```
eqs = [eq0, eq1, eq2, eq3, eq4]
for eq in eqs:
    pretty_print(eq, ' = 0')
```

$$2\left(\nu^2+\nu\right)f(r)^2 + (2\nu+1)f(r)\frac{\partial f}{\partial r} - \frac{1}{24}\left(\mu q^2 + 8\Lambda\right)f(r) + \frac{1}{2}f(r)\frac{\partial^2 f}{\partial r^2} = 0$$

$$\frac{1}{24}\mu q^2 - 2\left(\nu^2+\nu\right)f(r) - \nu\frac{\partial f}{\partial r} + \frac{1}{3}\Lambda = 0$$

$$-\frac{1}{12}\mu q^2 - 2(\nu+1)f(r) + \frac{1}{3}\Lambda - \frac{\partial f}{\partial r} = 0$$

$$\frac{1}{24}\lambda^2\mu q^2 + \frac{1}{3}\Lambda\lambda^2 - \frac{1}{2}\lambda^2\frac{\partial^2 f}{\partial r^2} - 2\left(\lambda^2\nu^2+\lambda^2+4\right)f(r) - \left(2\lambda^2\nu+\lambda^2\right)\frac{\partial f}{\partial r} = 0$$

$$-\frac{1}{8}\lambda^2\mu q^2 + 8(\nu+1)f(r) + 4\frac{\partial f}{\partial r} = 0$$

Let us solve `eq1` for $f(r)$:

In [40]:
```
sol_f = desolve(eq1.expr() == 0, f(r), ivar=r)
sol_f.expand()
```

Out[40]:

$$Ce^{(-2\,(\nu+1)r)} + \frac{\mu q^2}{48\,(\nu+1)\nu} + \frac{\Lambda}{6\,(\nu+1)\nu}$$

Hence, up to some rescaling the solution is of the type

$$f(r) = 1 - me^{-(2\nu+2)r},$$

where m is a constant. Hence we declare

In [41]:
```
var('m', domain='real')
fm(r) = 1 - m*exp(-(2*nu+2)*r)
fm
```

Out[41]: $r \mapsto -me^{(-2\,(\nu+1)r)} + 1$

and substitute this function for $f(r)$ in all the equations:

In [42]:
```
eq0m = eq0.expr().substitute_function(f, fm).simplify_full()
eq0m
```

Out[42]:

$$\frac{1}{24} \left(m\mu q^2 - 48\, m\nu^2 + 8\,\Lambda m - 48\, m\nu - \left(\mu q^2 - 48\,\nu^2 + 8\,\Lambda - 48\,\nu\right) e^{(2\,\nu r + 2\,r)}\right) e^{(-2\,\nu r - 2\,r)}$$

In [43]:
```
eq0m = (eq0m * exp(2*nu*r+2*r)).simplify_full()
eq0m
```

Out[43]:

$$\frac{1}{24}\, m\mu q^2 - 2\, m\nu^2 + \frac{1}{3}\,\Lambda m - 2\, m\nu - \frac{1}{24}\left(\mu q^2 - 48\,\nu^2 + 8\,\Lambda - 48\,\nu\right) e^{(2\,\nu r + 2\,r)}$$

In [44]:
```
eq1m = eq1.expr().substitute_function(f, fm).simplify_full()
eq1m
```

Out[44]:

$$\frac{1}{24}\,\mu q^2 - 2\,\nu^2 + \frac{1}{3}\,\Lambda - 2\,\nu$$

In [45]:
```
eq2m = eq2.expr().substitute_function(f, fm).simplify_full()
eq2m
```

Out[45]:

$$-\frac{1}{12}\,\mu q^2 + \frac{1}{3}\,\Lambda - 2\,\nu - 2$$

In [46]:
```
eq3m = eq3.expr().substitute_function(f, fm).simplify_full()
eq3m
```

Out[46]:

$$-\frac{1}{24}\left(48\,\lambda^2 m\nu - 48\,\left(\lambda^2+4\right)m - \left(\lambda^2\mu q^2 - 48\,\lambda^2\nu^2 + 8\,(\Lambda-6)\lambda^2 - 192\right)e^{(2\,\nu r+2\,r)}\right)e^{(-2\,\nu r-2\,r)}$$

In [47]:
```
eq3m = (eq3m * exp(2*nu*r+2*r)).simplify_full()
eq3m
```

Out[47]:

$$-2\,\lambda^2 m\nu + 2\,\left(\lambda^2+4\right)m + \frac{1}{24}\,\left(\lambda^2\mu q^2 - 48\,\lambda^2\nu^2 + 8\,(\Lambda-6)\lambda^2 - 192\right)e^{(2\,\nu r+2\,r)}$$

In [48]:
```
eq4m = eq4.expr().substitute_function(f, fm).simplify_full()
eq4m
```

Out[48]:

$$-\frac{1}{8}\,\lambda^2\mu q^2 + 8\,\nu + 8$$

In [49]:
```
eqs = [eq0m, eq1m, eq2m, eq3m, eq4m]
```

Solution for $\nu = 2$

In [50]:
```
neqs = [eq.subs(nu=2).simplify_full() for eq in eqs]
[eq == 0 for eq in neqs]
```

Out[50]:

$$\left[\frac{1}{24}\, m\mu q^2 + \frac{1}{3}(\Lambda - 36)m - \frac{1}{24}\left(\mu q^2 + 8\,\Lambda - 288\right)e^{(6\,r)} = 0, \frac{1}{24}\,\mu q^2 + \frac{1}{3}\,\Lambda - 12 = 0, -\frac{1}{12}\,\mu q^2 + \frac{1}{3}\,\Lambda - 6 = 0, -2\left(\lambda^2 - 4\right)m + \frac{1}{24}\left(\lambda^2 \mu q^2 + 8\,(\Lambda - 30)\lambda^2 - 192\right)e^{(6\,r)} = 0, -\frac{1}{8}\,\lambda^2 \mu q^2 + 24 = 0\right]$$

In [51]:
```
solve([eq == 0 for eq in neqs], lamb, mu, Lamb, q, m, r)
```

Out[51]:

$$\left[\left[\lambda = 2, \mu = \frac{48}{r_1^2}, \Lambda = 30, q = r_1, m = r_2, r = r_3\right], \left[\lambda = (-2)\,, \mu = \frac{48}{r_4^2}, \Lambda = 30, q = r_4, m = r_5, r = r_6\right]\right]$$

In the above solutions, r_i, with i an integer, stands for an arbitrary parameter. In particular, we notice that μ and q are related by $\mu q^2 = 48$ and that the value of m can be chosen arbitrarily.

Solution for $\nu = 4$

```
In [52]: neqs = [eq.subs(nu=4).simplify_full() for eq in eqs]
         [eq == 0 for eq in neqs]
```

Out[52]:

$$\left[\frac{1}{24}\, m\mu q^2 + \frac{1}{3}\,(\Lambda - 120)m - \frac{1}{24}\,\left(\mu q^2 + 8\,\Lambda - 960\right)e^{(10\,r)} = 0, \frac{1}{24}\,\mu q^2 + \frac{1}{3}\,\Lambda - 40 = 0, -\frac{1}{12}\,\mu q^2 + \frac{1}{3}\,\Lambda - 10 = 0, -2\,\left(3\,\lambda^2 - 4\right)m + \frac{1}{24}\,\left(\lambda^2\mu q^2 + 8\,(\Lambda - 102)\lambda^2 - 192\right)e^{(10\,r)} = 0, -\frac{1}{8}\,\lambda^2\mu q^2 + 40 = 0\right]$$

```
In [53]: solve([eq == 0 for eq in neqs], lamb, mu, Lamb, q, m, r)
```

Out[53]:

$$\left[\left[\lambda = \frac{2}{3}\,\sqrt{3}, \mu = \frac{240}{r_7^2}, \Lambda = 90, q = r_7, m = r_8, r = r_9\right], \left[\lambda = -\frac{2}{3}\,\sqrt{3}, \mu = \frac{240}{r_{10}^2}, \Lambda = 90, q = r_{10}, m = r_{11}, r = r_{12}\right]\right]$$

As above, r_i, with i an integer, stands for an arbitrary parameter. The constants μ and q are now related by $\mu q^2 = 240$ and the value of m is still arbitrary.

Graphical example: the 2-sphere



Stereographic coordinates on the 2-sphere

Two charts:

- X_1: $\mathbb{S}^2 \setminus \{N\} \to \mathbb{R}^2$
- X_2: $\mathbb{S}^2 \setminus \{S\} \to \mathbb{R}^2$

← picture drawn with `Chart.plot()`

See the worksheet at
http://sagemanifolds.obspm.fr/examples/html/SM_sphere_S2.html

Graphical example: the 2-sphere



Vector frame associated with the stereographic coordinates (x, y) from the North pole

- $\frac{\partial}{\partial x}$
- $\frac{\partial}{\partial y}$

← picture drawn with `VectorField.plot()`

Charts on Schwarzschild spacetime
The Carter-Penrose diagram



Two charts of standard Schwarzschild-Droste coordinates (t, r, θ, φ) plotted in terms of compactified coordinates $(\tilde{T}, \tilde{X}, \theta, \varphi)$

See the worksheet at

http://sagemanifolds.obspm.fr/examples/html/SM_Carter-Penrose_diag.html

- yellow: infalling shell of radiation
- green: radial null geodesics
- red: trapping horizon
- black: event horizon

See the worksheet at
http://nbviewer.jupyter.org/github/egourgoulhon/BHLectures/blob/master/sage/Vaidya.ipynb



Conclusion and perspectives

- **SageManifolds** is a work in progress
  $\sim$ 64,000 lines of Python code up to now (including comments and doctests)
- A preliminary version (v0.9) is freely available (GPL) at
  `http://sagemanifolds.obspm.fr/`
  with the following features:
  - maps between manifolds, pullback operator
  - curves in manifolds
  - standard tensor calculus (tensor product, contraction, symmetrization, etc.), even on non-parallelizable manifolds
  - all monoterm tensor symmetries
  - exterior calculus (wedge product, exterior derivative, Hodge duality)
  - Lie derivatives of tensor fields
  - affine connections, curvature, torsion
  - pseudo-Riemannian metrics, Weyl tensor
  - some plotting capabilities (charts, points, curves, vector fields)
  - parallelization (on tensor components) of CPU demanding computations, via the Python library `multiprocessing`

Conclusion and perspectives

SageManifolds is aimed to be fully integrated into SageMath

Ongoing review process by the SageMath community: cf. the metaticket
https://trac.sagemath.org/ticket/18528
Some parts are already included in SageMath 7.2 (more in the next 7.3)

Meanwhile, one can either

- install it atop SageMath via a simple script
- use it online on the SageMathCloud (where it is installed system-wide): just open a free account at https://cloud.sagemath.com/

Want to join the project or simply to stay tuned?

visit http://sagemanifolds.obspm.fr/
(download, documentation, example worksheets, mailing list)